

REPORT 1968-2

REPORT  
by  
THE OHIO STATE UNIVERSITY RESEARCH FOUNDATION  
COLUMBUS, OHIO 43212

Sponsor

Bureau of Naval Weapons  
Washington, D. C.

Contract Number

NOw 65-0329-d

Investigation of

Reactive-Wall Reflectors for 9mm Microwave  
Operation

Subject of Report

Computer Optimization of a Corner Reflector  
for Echo Enhancement with Circular Polarization

Submitted by

Jack H. Richmond  
Antenna Laboratory  
Department of Electrical Engineering

Date

22 September 1965

Release to Defense Center (ASTIA) and to OTS, Dept. of  
Commerce, Washington, D. C., without restriction. The  
contractor imposes no limitation or restriction upon the  
distribution of this document by the Government to both  
Government and non-Government recipients.

CLEARINGHOUSE  
FOR FEDERAL SCIENTIFIC  
TECHNICAL INFORMATION

Harlequin

Microfilm

1.00

0.50

19

ARCHIVE COPY

T. D. C.

SEP 15 1965

## ABSTRACT

When a circularly polarized wave is incident on a trihedral corner reflector, the backscattered wave is circularly polarized with the opposite sense of rotation. However, by coating one of the conducting surfaces of the reflector with dielectric layers, it is possible to obtain a circularly polarized backscatter wave having the same sense of rotation as the incident wave. With the aid of a digital computer, the thicknesses of the dielectric layers can be optimized in an efficient manner to obtain the best performance over the pertinent range of incidence angles. This report includes a computer program for this purpose and describes an optimized four-layer reactive wall.

USE  
FILE AND  
ATION

19

MPV

## TABLE OF CONTENTS

	Page
I. INTRODUCTION	1
II. COMPUTER OPTIMIZATION	1
III. AN EXAMPLE OF REACTIVE WALL OPTIMIZATION	4
IV. CONCLUSIONS	9
REFERENCES	10
APPENDIX - DIGITAL COMPUTER PROGRAM	11

## COMPUTER OPTIMIZATION OF A CORNER REFLECTOR FOR ECHO ENHANCEMENT WITH CIRCULAR POLARIZATION

### I. INTRODUCTION

This report considers the design of a reactive wall consisting of a dielectric multilayer on a conducting plane. This reactive wall is to be used as one face of a trihedral corner reflector in which the two remaining faces are uncoated conducting planes. When a right-circularly polarized wave is incident on the corner reflector, it is desired in some applications that the backscattered wave also be right-circularly polarized.

When there are only one or two dielectric layers, the appropriate thicknesses of the layers can be determined by graphical constructions on the Smith chart, using a method devised by Kennaugh[1]. However, Kennaugh and Chung[2] have shown that even the carefully designed reactive walls have rather poor characteristics when there are only one, two or three layers.

Although the Smith chart technique yields a suitable design for any given angle of incidence, it does not lead to a design which is optimum over a range of angles of incidence. This report describes a method for obtaining such an optimum design with the aid of an automatic digital computer. The technique is similar to that developed by Kotik and Cope[3] for the optimization of multilayer radome walls.

The following sections describe the technique and present a computer program and typical results for an optimized reactive wall having four layers.

### II. COMPUTER OPTIMIZATION

Consider a harmonic plane wave to be incident on a perfectly conducting plane coated with a dielectric multilayer as shown in Fig. 1. The reflection coefficients for the TE and TM cases can be calculated with the aid of a computer program included in Report 1968-1[4]. If the dielectric layers are lossless, the TE and TM reflection coefficients are given by




Fig. 1. A plane wave incident on a reactive wall consisting of four dielectric layers on a conducting plane.

$$(1) \quad R_e = e^{j\delta}$$

$$(2) \quad R_m = e^{j\delta'}$$

That is, total reflection occurs and the reflection phase angles are denoted by  $\delta$  and  $\delta'$ . Let  $R_e$  denote the ratio of the electric field intensities of the reflected and incident plane waves in the TE case (perpendicular polarization) and let  $R_m$  denote the ratio of the magnetic field intensities of the reflected and incident plane waves in the TM case (parallel polarization). If the incident wave has right-circular polarization, the reflected wave will in general have both right- and left-circularly polarized components. In the present application we are interested in the "same-sense power reflection coefficient" defined by the ratio of the power density of the right-circularly polarized component of the reflected wave and the power density of the right-circularly polarized incident wave. This same-sense power reflection coefficient, denoted by  $R$ , is given by

$$(3) \quad R = \cos^2 \phi$$

where

$$(4) \quad \phi = (\delta - \delta')/2 \quad .$$

It may be noted that the reflection phase difference  $(\delta - \delta')$  is equal to the phase angle of the ratio of the complex reflection coefficients:

$$(5) \quad \delta - \delta' = \text{Phase } (R_e/R_m) \quad .$$

If the dielectric layers are dissipative, the same-sense power reflection coefficient is given by

$$(6) \quad R = 0.25 |R_e + R_m|^2 \quad .$$

Suppose that the dielectric constants  $\epsilon_1$ ,  $\epsilon_2$ ,  $\epsilon_3$ , and  $\epsilon_4$  of the four layers are specified and we are to design the thicknesses  $d_1$ ,  $d_2$ ,  $d_3$ , and  $d_4$  to maximize the same-sense power reflection coefficient  $R$  over a given range of incidence angles. (In the case of the trihedral corner reflector, Kennaugh[1] has shown that the pertinent range of incidence angles is from  $\theta = 35^\circ$  to  $\theta = 75^\circ$ .) To be more specific, suppose that the layer thicknesses are to be designed to maximize the minimum value of  $R$  over this range of incidence angles. That is, the "figure of merit" of any given design is taken to be the minimum value of  $R$  in the range from  $\theta = 35^\circ$  to  $75^\circ$ , and this quantity is denoted by  $R_m$ .  $R_m$  is a function of the thicknesses  $d_1$ ,  $d_2$ ,  $d_3$ , and  $d_4$ . It is also a function of the frequency and the dielectric constants of the layers, but these are assumed to be specified and the permeability of each layer is taken to be that of free space.

The most straightforward (but impractical) method for determining the layer thicknesses is to program the computer to calculate and print out  $R$  versus incidence angle for a large number of designs, covering a given range in each of the thicknesses. A study of the resulting data would indicate the region of the best design. However, this simple method requires the calculation and inspection of the results for ten thousand different designs obtained by taking ten increments in each of the four thicknesses. A more sophisticated approach can lead to the optimum design with much greater efficiency.

This digital optimization technique proceeds as follows. An initial design is specified in the form of a set of numerical values for the four thicknesses. The computer is programmed to calculate the figure of merit  $R_m$  for this initial design and for four slightly different designs involving small changes in the layer thicknesses. Based on this information, the computer determines the partial derivatives of  $R_m$  with respect to the thickness of each layer. It then arrives at a new and better design (having a larger value of  $R_m$ ) by changing the thickness of each layer by an amount proportional to the partial derivative of  $R_m$  with respect to that thickness. For this new design,  $R_m$  and the partial derivatives are again calculated in order to arrive at a still better design. After repeating this procedure several times, the computer soon arrives at the optimum design. In a typical case, the computer investigates 40 different designs in order to find the optimum. In finding the partial derivatives, it investigates four other designs in the vicinity of each of these 40 basic ones, making a total of 200. If this is compared with the 10,000 designs which must be investigated in the method described previously, the tremendous advantage of the computer optimization technique becomes apparent.

The computer programs for the optimization process are presented in the Appendix.

### III. AN EXAMPLE OF REACTIVE WALL OPTIMIZATION

As an example, consider the optimization of a four-layer reactive wall having lossless layers with relative dielectric constants given by  $\epsilon_1 = 1$ ,  $\epsilon_2 = 9.6$ ,  $\epsilon_3 = 1$ , and  $\epsilon_4 = 9.6$ . The initial design is specified by the thicknesses  $d_1 = 0.05\lambda$ ,  $d_2 = 0.1\lambda$ ,  $d_3 = 0.05\lambda$ , and  $d_4 = 0.1\lambda$  where  $\lambda$  denotes the free-space wavelength. The digital computer calculates the same-sense power reflection coefficient  $R$  for this initial design at angles of incidence from  $\theta = 35^\circ$  to  $75^\circ$  in increments of  $5^\circ$ . It also calculates  $R$  versus  $\theta$  for four slightly different designs in which the thickness of one layer is increased by  $0.001\lambda$ . Table I lists the thicknesses of the layers for each of these five cases.

---

\* The initial design may be based on experience or on the results of a graphical solution on the Smith chart. To a certain extent, the final results will be the same regardless of the starting point, but the computational costs are reduced by selecting a starting point close to the optimum design.

TABLE I  
LIST OF LAYER THICKNESSES  
 $d/\lambda$

<u>Layer No.</u>	<u>Case 1</u>	<u>Case 2</u>	<u>Case 3</u>	<u>Case 4</u>	<u>Case 5</u>
1	0.050	0.051	0.050	0.050	0.050
2	0.100	0.100	0.101	0.100	0.100
3	0.050	0.050	0.050	0.051	0.050
4	0.100	0.100	0.100	0.100	0.101

Table II lists the power reflection coefficient  $R$  versus angle of incidence for these five cases.

TABLE II  
SAME-SENSE POWER REFLECTION COEFFICIENT

<u><math>\theta</math></u>	<u>Case 1</u>	<u>Case 2</u>	<u>Case 3</u>	<u>Case 4</u>	<u>Case 5</u>
35°	.3600	.3425	.3256	.3456	.3378
40	.7516	.7297	.7045	.7361	.7218
45	.9948	.9913	.9850	.9930	.9890
50	.9316	.9374	.9479	.9332	.9428
55	.7947	.7987	.8120	.7935	.8076
60	.6913	.6925	.7051	.6881	.7030
65	.6320	.6313	.6436	.6282	.6433
70	.6150	.6134	.6261	.6115	.6269
75	.6467	.6448	.6583	.6441	.6596

In each case the minimum reflection coefficient occurs at  $\theta = 35^\circ$ . The figure of merit for each case is  $R_m = 0.3600, 0.3425, 0.3256, 0.3456$ , and  $0.3378$ . The best design out of these five cases is Case I, since it has the highest figure of merit. Since  $R_m$  decreases when the thickness of any layer is increased slightly, it is obvious that a better design can be obtained by decreasing the thicknesses of all the layers, unless by chance the initial design (Case 1) is optimum.

From Table II it is seen that a change in the thickness of layer 2 has the greatest effect on  $R_m$ , and a change in layer 3 has the least



effect. Thus, in arriving at a better design, we should decrease  $d_2$  by a relatively large amount and layer 3 by a relatively small amount.

If the figures of merit for the five cases are denoted by  $R_1$ ,  $R_2$ ,  $R_3$ ,  $R_4$ , and  $R_5$ , the partial derivatives of  $R_m$  with respect to the layer thicknesses are given approximately by

$$(7) \quad \frac{\partial R_m}{\partial d_1} = (R_2 - R_1)/s$$

$$(8) \quad \frac{\partial R_m}{\partial d_2} = (R_3 - R_1)/s$$

and so forth, where  $s$  represents the increment in thickness which is  $0.001\lambda$  in this example. In order to obtain an improved design, all four layers in the initial design are now changed in thickness by amounts proportional to the partial derivatives. In this manner the computer arrives at the first improved design with thicknesses  $d_1 = 0.0498\lambda$ ,  $d_2 = 0.0996\lambda$ ,  $d_3 = 0.0498\lambda$ , and  $d_4 = 0.0997\lambda$ . The minimum reflection coefficient over the specified range of incidence angles is found to be  $R_m = 0.3867$ . This represents a definite improvement over the initial design, since the figure of merit has increased from 0.3600 to 0.3867.

After 40 cycles of this procedure, the computer arrived at a design which is very close to the optimum. This optimum design is specified in Fig. 2 which also shows the same-sense power reflection coefficient as a function of the angle of incidence. This reflection coefficient exceeds 0.965 over the specified range of incidence angles. Thus, the figure of merit of the final design is 0.965. The entire optimization procedure required only 3 minutes on an IBM 7094 computer, including a second run in which the thickness increment was reduced to  $0.0002\lambda$  and the incidence angle increment was reduced to  $2^\circ$ . The initial design for the second run was taken to be the best design obtained from the first run.

Figure 2 also shows the same-sense power reflection coefficient for the more practical case where the losses in the dielectric layers are taken into account. The loss tangent was taken to be zero for layers 1 and 3 and 0.001 for layers 2 and 4. The optimum layer thicknesses were found to be  $0.0426\lambda$ ,  $0.0897\lambda$ ,  $0.0464\lambda$ , and  $0.0980\lambda$  in this case, and the minimum power reflection coefficient is 0.929. These thicknesses differ very little from those of the optimized lossless multilayer which are listed in Fig. 2.




Fig. 2. Same-sense power reflection coefficient versus angle of incidence for a circularly-polarized wave incident on a four-layer reactive wall with optimized thicknesses.

The optimization procedure described above was carried out for three reactive wall designs using lossless layers with different dielectric constants. The dielectric constants and the optimum layer thicknesses for these cases are listed in Table III.

TABLE III  
OPTIMIZED REACTIVE WALL DESIGNS  
WITH LOSSLESS LAYERS  
Design Frequency: 33.2 Gc

<u>Dielectric Constants</u>			
<u>Layer No.</u>	<u>Case 1</u>	<u>Case 2</u>	<u>Case 3</u>
1	1.0	1.0	1.0
2	6.0	8.0	9.6
3	1.0	1.0	1.0
4	6.0	8.0	9.6

<u>Thicknesses in Inches</u>			
<u>Layer No.</u>	<u>Case 1</u>	<u>Case 2</u>	<u>Case 3</u>
1	0.02752	0.01810	0.01512
2	0.04157	0.03547	0.03182
3	0.02652	0.01860	0.01646
4	0.03110	0.03578	0.03477

Each of these three designs was found to have excellent characteristics in the frequency range from 33.0 to 33.4 Gc and the incidence angle range from 35° to 75°. Of these three designs, Case 1 is probably the most promising because the allowable tolerances in the layer thicknesses are undoubtedly less critical than in the other two cases. Table IV lists the reflection coefficients for Case 1 as a function of frequency and angle of incidence.

TABLE IV  
SAME-SENSE POWER REFLECTION COEFFICIENT  
FOR CASE 1

Angle of Incidence	33.0 Gc	33.1 Gc	33.2 Gc	33.3 Gc	33.4 Gc
35°	.81	.81	.81	.81	.80
37	.88	.88	.89	.89	.89
39	.93	.93	.94	.94	.94
41	.96	.97	.97	.97	.98
43	.98	.98	.99	.99	.99
45	.99	.99	.99	.99	.99
47	.99	.99	.99	.99	.99
49	.99	.99	.99	.99	.99
51	.99	.99	.99	.99	.99
53	.99	.99	.99	.99	.99
55	.99	.99	.99	.99	.99
57	.99	.99	.99	.99	.99
59	.99	.99	.99	.99	.99
61	.98	.99	.99	.99	.99
63	.96	.98	.99	.99	.99
65	.94	.97	.99	.99	.99
67	.91	.95	.98	.99	.99
69	.86	.92	.96	.99	.99
71	.81	.87	.93	.97	.99
73	.74	.82	.88	.94	.98
75	.67	.75	.83	.90	.96

#### IV. CONCLUSIONS

In a reactive wall for a corner reflector, the thicknesses of the dielectric layers can be designed most efficiently with a computer optimization technique. Starting with a specified initial design, an automatic digital computer is programmed to take small increments in the thicknesses in such a manner as to move in the direction of maximum rate of improvement. Excellent new designs are obtained in this manner at low cost. An optimized four-layer wall is described which has a same-sense power reflection coefficient exceeding 0.965 for all angles of incidence in the range of interest.

Future research along these lines should investigate the effects of manufacturing tolerances in the thicknesses of the various layers. An improved design should be attempted which will have a larger bandwidth. Experimental verification should be carried out for one or more of the optimized corner reflectors.

#### REFERENCES

1. Kennaugh, E. M., "Polarization Properties of Corner Reflectors With Modified Walls," Report 612-6, 1 January 1957, Antenna Laboratory, The Ohio State University Research Foundation; prepared under Contract AF 30(635)-2811, Rome Air Development Center, Griffiss Air Force Base, New York. (AD 430 140).
2. Kennaugh, E. M., and C. Chung, "Design and Theoretical Performance of Reactive-Wall Corner Reflectors," Report 1790-1, 13 January 1965, Antenna Laboratory, The Ohio State University Research Foundation; prepared under Contract NOW-64-0215-d, Department of the Navy, Bureau of Naval Weapons, Washington, D. C. (AD 458 004).
3. Kotik, J. and D. Cope, "Digital-Optimization of Radome Walls," Technical Research Group, Proceedings of the OSU-WADD Symposium on Electromagnetic Windows, June 1960, Vol. 1, pp. 220-234.
4. Richmond, J. H., "Efficient Recursive Solutions for Plane and Cylindrical Multilayers," Report 1968-1, 10 August 1965, Antenna Laboratory, The Ohio State University Research Foundation; prepared under Contract NOW-65-0329-d, Bureau of Naval Weapons, Washington, D. C.

## APPENDIX DIGITAL COMPUTER PROGRAM

Figure 3 shows a computer program which has been used successfully for optimizing the thicknesses of the dielectric layers in a reactive wall. The program is written in the computer language known as "Scatran". In its present form the program is set up for a four-layer wall with lossless layers, but it can readily be modified for any number of lossless or dissipative layers.

The input data symbols are defined as follows:

KKK	= Number of angles of incidence
MMM	= Maximum number of steps to take in seeking the optimum design
E(I)	= Relative dielectric constant of layer i
DA, DB, DC, DD	= Initial thickness, in free-space wavelengths, of layers 1, 2, 3, and 4
DDD	= Thickness increment, in free-space wavelengths
THETA	= Initial angle of incidence, degrees
DTHET	= Increment in angle of incidence, degrees

The core of the program, between statements S10 and S100, is a specialized version of a program described in Report 1968-1[4] for the reflection coefficients of a plane dielectric multilayer. In order to increase the speed and the efficiency of the calculations, the general program given in Reference 4 is specialized here for the case where the dielectric layers are lossless and have the same permeability as free space. Furthermore, the general program gives solutions for the reflection and transmission coefficients of a multilayer with and without a conducting plane. The program is simplified considerably here where the multilayer without a conducting plane is of no interest.

The program includes provisions for terminating the run if the thickness of any layer goes negative. The run also terminates if the number of "unsuccessful steps" exceeds five. A step is considered to be unsuccessful if simultaneous increments in all four thicknesses leads to a new design which has a smaller figure of merit than that of the previous design.

Just before statement S100, the quantity DEL denotes the reflection phase difference ( $\delta - \delta'$ ) which appears in Eqs. (4) and (5), and RRR(KK) is the same-sense power reflection coefficient which is denoted by R in Eq. (3). Just after statement S100, the program directs the computer to

print out the list of layer thicknesses ( $D_1$ ,  $D_2$ ,  $D_3$ , and  $D_4$ ) and the power reflection coefficient  $RRR(I)$  versus angle of incidence.

The quantities  $R_1$ ,  $R_2$ ,  $R_3$ ,  $R_4$ , and  $R_5$  which appear below statement S110 represent the figures of merit denoted by  $R_1$ ,  $R_2$ ,  $R_3$ , etc. in Eqs. (7) and (8) for Cases 1 through 5. These figures of merit are used to calculate the partial derivatives with respect to the thickness of each layer. Just above statement S150, the thickness of each layer is incremented by an amount proportional to the partial derivative of the figure of merit. The quantity  $D$  which is calculated between statements S110 and S150 is defined by

$$(9) \quad D = |R_2 - R_1| + |R_3 - R_1| + |R_4 - R_1| + |R_5 - R_1| .$$

If it were important to take steps of uniform length in moving toward the optimum design, it would be correct to define  $D$  by

$$(10) \quad D = \sqrt{(R_2 - R_1)^2 + (R_3 - R_1)^2 + (R_4 - R_1)^2 + (R_5 - R_1)^2} .$$

However, the definition used in Eq. (9) and in the computer program has proven to be satisfactory.

At the end of the computer program in Fig. 3, a typical set of input data are shown.

Figure 4 shows the computer program used for optimizing a four-layer reactive wall with dissipative layers. The list of loss tangents, denoted by  $TD(1)$ ,  $TD(2)$ ,  $TD(3)$ , and  $TD(4)$ , must be given along with the other input data described previously for the program in Fig. 3.

```

***      RUN
***      SCATRAM
COMPLEX(C1,AEC,BEC,AMC,BMC)EXP1,EXP11,EXP,AECP,REC,AMCP,
RMC,RSR)-
DIMENSION(E(100),Z(100),G(100),D1(10),D2(10),D3(10),D4(10),
RR(300),R(30),RRR(30))-
FIRST   READ INPUT,8,(KKK,MMM)-
        N=4-
        NN=N+1-
        READ INPUT,7,(E(1),E(2),E(3),E(4))-
        READ INPUT,7,(DA,DB,DC,DD,DDD)-
        READ INPUT,7,(THETA,DTHET)-
        TPI=6.2831853-
        E(NN)=1.-
        RR(0)=.0-
        WRITE OUTPUT,2,(E(1),E(2),E(3),E(4),E(5),THETA)-
        WRITE OUTPUT,2-
        L=1-
        D1(1)=DA-
        D2(1)=DB-
        D3(1)=DC-
        D4(1)=DD-
        DO THROUGH(S150),MM=1,1,MM,LE,MMM-
        DO THROUGH(S20),I=2,1,1,LE,5-
        D1(1)=D1(1)-
        D2(1)=D2(1)-
        D3(1)=D3(1)-
S20      D4(1)=D4(1)-
        D1(2)=D1(1)+DDD-
        D2(3)=D2(1)+DDD-
        D3(4)=D3(1)+DDD-
        D4(5)=D4(1)+DDD-
        PROVIDED(D1(1).L.0.),TRANSFER TO (S160)-
        PROVIDED(D2(1).L.0.),TRANSFER TO (S160)-
        PROVIDED(D3(1).L.0.),TRANSFER TO (S160)-
        PROVIDED(D4(1).L.0.),TRANSFER TO (S160)-
        DO THROUGH(S110),LL=1,1,LL,LE,5-
        Z(1)=TPI*D1(LL)-
        Z(2)=Z(1)+TPI*D2(LL)-
        Z(3)=Z(2)+TPI*D3(LL)-
        Z(4)=Z(3)+TPI*D4(LL)-
        THET=THETA-
        DO THROUGH(S100),KK=1,1,KK,LE,KKK-
        TH=.01745329*THET-
        SS=SIN.(TH)*SIN.(TH)-
        CC=COS.(TH)-
        DO THROUGH(S10),I=1,1,1,LE,N-
S10      G(I)=SQRT.(E(I)-SS)-
        G(NN)=CC-
        AEC=1.-
        BEC=-1.-
        AMC=1.-
        BMC=1.-
        DO THROUGH(S50),II=2,1,II,LE,NN-

```

Fig. 3. Digital computer program for optimizing the thicknesses of a reactive wall with four lossless layers.



```

      I=I1-I-
      F=G(I)/G(I1)-
      YI=G(I)*Z(I)-
      YI1=G(I1)*Z(I)-
      FP=1.+F-
      FM=1.-F-
      EXP1=COS.(YI)+.I.SIN.(YI)-
      EXP11=COS.(YI1)+.I.SIN.(YI1)-
      AECP=AEC-
      AMCP=AMC-
      AEC=(.5/EXP11)*(AECP*EXP1*FP+BEC*FM/EXP1)-
      BEC=.5*EXP11*(AECP*EXP1*FM+BEC*FP/EXP1)-
      F=F*E(I1)/E(I)-
      FP=1.+F-
      FM=1.-F-
      AMC=(.5/EXP11)*(AMCP*EXP1*FP+BMC*FM/EXP1)-
S50    BMC=.5*EXP11*(AMCP*EXP1*FM+BMC*FP/EXP1)-
      RSR=BEC*AMC/(AEC*BMC)-
      DEL=FATAN2.(.IMAG.RSR,.REAL.RSR)-
S100   RRR(KK)=.5*(1.+COS.(DEL))-
      THET=THET+DTHET-
      WRITE OUTPUT,2,(D1(LL),D2(LL),D3(LL),D4(LL))-
      WRITE OUTPUT,2,((RRR(I),I=1,1,1,LE,KKK))-
      R(LL)=RRR(1)-
      DO THROUGH(S102),I=2,1,1,LE,KKK-
S102   PROVIDED(RRR(I),L,R(LL)),R(LL)=RRR(I)-
S110   CONTINUE-
      R1=R(1)-
      R2=R(2)-
      R3=R(3)-
      R4=R(4)-
      R5=R(5)-
      D=.ABS.(R2-R1)+.ABS.(R3-R1)+.ABS.(R4-R1)+.ABS.(R5-R1)-
      RR(MM)=R1-
      WRITE OUTPUT,2-
      WRITE OUTPUT,2,(D1(1),D2(1),D3(1),D4(1))-
      WRITE OUTPUT,2,(R1,R2,R3,R4,R5)-
      WRITE OUTPUT,2-
      PROVIDED(R1,L,RR(MM-1)),L=L+1-
      PROVIDED(L,G,5),TRANSFER TO(S160)-
      D1(1)=D1(1)+DDD*(R2-R1)/D-
      D2(1)=D2(1)+DDD*(R3-R1)/D-
      D3(1)=D3(1)+DDD*(R4-R1)/D-
      D4(1)=D4(1)+DDD*(R5-R1)/D-
S150   WRITE OUTPUT,2-
S160   CONTINUE-
      END PROGRAM(FIRST)-
***    DATA
      9      50
          1.0      9.6      1.0      9.6
          .05      .10      .05      .10      .001
          35.0     5.0

```

Fig. 3. Digital computer program for optimizing the thicknesses of a reactive wall with four lossless layers. (cont).

```

***      RUN
***      SCATRAM
COMPLEX(CI,AEC,BEC,AMC,BMC,EXPI,EXPII,EXP,AECP,REC,AMCP,
RMC,EC,G,CSQRTL,CEXPL,F,YI,YII,FP,FM)-
DIMENSION(E(100),Z(100),G(100),D1(10),D2(10),D3(10),D4(10),
RR(300),R(30),RRR(30),TD(10),EC(10))-
FIRST    READ INPUT,8,(KKK,MMM)-
          N=4-
          NN=N+1-
          READ INPUT,7,(E(1),E(2),E(3),E(4))-
          READ INPUT,7,(TD(1),TD(2),TD(3),TD(4))-
          READ INPUT,7,(DA,DB,DC,DD,DDD)-
          READ INPUT,7,(THETA,DTHET)-
          TPI=6.2831853-
          CI=.1.1.-
          E(NN)=1.-
          EC(NN)=1.-
          RR(0)=.0-
          WRITE OUTPUT,2,(E(1),E(2),E(3),E(4),E(5),THETA)-
          WRITE OUTPUT,2,(TD(1),TD(2),TD(3),TD(4))-
          WRITE OUTPUT,2-
          L=1-
          D1(1)=DA-
          D2(1)=DB-
          D3(1)=DC-
          D4(1)=DD-
          DO THROUGH(S5),I=1,1,1,LE,N-
S5        EC(I)=E(I)*(1.-CI*TD(I))-
          DO THROUGH(S150),MM=1,1,MM,LE,MMM-
          DO THROUGH(S20),I=2,1,1,LE,5-
          D1(I)=D1(I)-
          D2(I)=D2(I)-
          D3(I)=D3(I)-
S20       D4(I)=D4(I)-
          D1(2)=D1(1)+DDD-
          D2(3)=D2(1)+DDD-
          D3(4)=D3(1)+DDD-
          D4(5)=D4(1)+DDD-
          PROVIDED(D1(1).L.0.),TRANSFER TO (S160)-
          PROVIDED(D2(1).L.0.),TRANSFER TO (S160)-
          PROVIDED(D3(1).L.0.),TRANSFER TO (S160)-
          PROVIDED(D4(1).L.0.),TRANSFER TO (S160)-
          DO THROUGH(S110),LL=1,1,LL,LE,5-
          Z(1)=TPI*D1(LL)-
          Z(2)=Z(1)+TPI*D2(LL)-
          Z(3)=Z(2)+TPI*D3(LL)-
          Z(4)=Z(3)+TPI*D4(LL)-
          THET=THETA-
          DO THROUGH(S100),KK=1,1,KK,LE,KKK-
          TH=.01745329*THET-
          SS=SIN.(TH)*SIN.(TH)-
          CC=COS.(TH)-
S10       DO THROUGH(S10),I=1,1,1,LE,N-
          G(I)=CI*CSQRTL.(EC(I)-SS)-
          G(NN)=CI*CC-
          AEC=1.-
          BEC=-1.-

```

Fig. 4. Digital computer program for optimizing the thicknesses of a reactive wall with four lossy layers.

```

      AMC=1.-
      BMC=1.-
      DO THROUGH(S50),I=2,1,11.LE.NN-
      I=I-1-
      F=G(I)/G(11)-
      YI=G(I)*Z(I)-
      YII=G(11)*Z(I)-
      FP=1.+F-
      FM=1.-F-
      EXP1=CEXPL.(YI)-
      EXP11=CEXPL.(YII)-
      AECP=AEC-
      AMCP=AMC-
      AEC=(.5/EXP11)*(AECP*EXP1*FP+BEC*FM/EXP1)-
      BEC=.5*EXP11*(AECP*EXP1*FM+BEC*FP/EXP1)-
      F=F*E(11)/E(1)-
      FP=1.+F-
      FM=1.-F-
      AMC=(.5/EXP11)*(AMCP*EXP1*FP+BMC*FM/EXP1)-
S50    BMC=.5*EXP11*(AMCP*EXP1*FM+BMC*FP/EXP1)-
      REC=BEC/AEC-
      RMC=BMC/AMC-
      RMAG=.ABS.(REC+RMC)-
S100   RRR(KK)=.25*RMAG*RMAG-
      THET=THET+DTHET-
      WRITE OUTPUT,2,(D1(LL),D2(LL),D3(LL),D4(LL))-
      WRITE OUTPUT,2,((RRR(I),I=1,1,1.LE.KKK))-
      R(LL)=RRR(I)-
      DO THROUGH(S102),I=2,1,1.LE.KKK-
S102   PROVIDED(RRR(I).L.R(LL)),R(LL)=RRR(I)-
S110   CONTINUE-
      R1=R(1)-
      R2=R(2)-
      R3=R(3)-
      R4=R(4)-
      R5=R(5)-
      D=.ABS.(R2-R1)+.ABS.(R3-R1)+.ABS.(R4-R1)+.ABS.(R5-R1)-
      RR(MM)=R1-
      WRITE OUTPUT,2-
      WRITE OUTPUT,2,(D1(1),D2(1),D3(1),D4(1))-
      WRITE OUTPUT,2,(R1,R2,R3,R4,R5)-
      WRITE OUTPUT,2-
      PROVIDED(R1.L.RR(MM-1)),L=L+1-
      PROVIDED(L.G.5),TRANSFER TO(S160)-
      D1(1)=D1(1)+DDD*(R2-R1)/D-
      D2(1)=D2(1)+DDD*(R3-R1)/D-
      D3(1)=D3(1)+DDD*(R4-R1)/D-
      D4(1)=D4(1)+DDD*(R5-R1)/D-
S150   WRITE OUTPUT,2-
S160   CONTINUE-
      END PROGRAM(FIRST)-
      DATA
***
      9      20
          1.0          9.6          1.0          9.6
          .000          .001          .000          .001
      .042525      .089520      .046316      .097800      .0002
          35.0          5.0

```

Fig. 4. Digital computer program for optimizing the thicknesses of a reactive wall with four lossy layers. (cont).